

Techniken für weniger Stack-Tiefe

M. Anton Ertl
TU Wien

Was tun, wenn zuviele Daten herumschwirren?

- Mehrere Stacks
- Locals
- Implizite Parameter
- Aufteilung auf Tasks
- Stufenweise Ausführung

Implizite Parameter

- Zähler und Grenze bei Zählschleifen
- Globale Variablen, z.B. `base`: Nachteile
- Kontext, z.B. `base-execute`, `>string-execute`

Beispiel: Vektorverarbeitung

```
: innerproduct ( a[row][*] b[*][column] -- int)
  0 row-size 0 do
    >r over @ over @ * r> + >r
    swap cell+ swap row-byte-size +
    r>
  loop
  >r 2drop r> ;

: faxpy ( ra f-addr-x nstride-x f-addr-y nstride-y ucount -- )
  \ vy=ra*vx+vy
  >r swap 2swap swap r> 0 ?D0
    fdup dup f@ f* over + 2swap dup f@ f+ dup f!
    over + 2swap
  LOOP
  2drop 2drop fdrop ;
```

Aufteilung auf Tasks

```
: v@ ( f-addr nstride ucount -- )
  0 ?do
    over f@ fput
    tuck + loop
  endput 2drop ;

: vf* ( ra -- )
  begin fget? while
    fover f* fput repeat
  fdrop ;

: v+! ( f-addr nstride -- )
  begin fget? while
    over f@ f+ over f!
    tuck + repeat
  2drop ;

: faxpy ( ra f-addr-x nstride-x f-addr-y nstride-y ucount -- )
  rot rot 2>r ['] v@ xxx|
  ['] vf* f|
  2r> v+! ;
```

Stufenweise Ausführung

```
: gen-innerproduct ( a[row][*] -- xt )
\ xt is of type ( b[*][column] -- n )
\ this would be a candidate for using ]] ... [[
>r :noname r>
0 POSTPONE literal POSTPONE SWAP
row-size 0 do
  POSTPONE dup POSTPONE @
  dup @ POSTPONE literal POSTPONE * POSTPONE under+
  POSTPONE cell+ row-byte-size +
loop
drop
POSTPONE drop POSTPONE ;

... ( a ) gen-innerproduct ... ( b xt ) execute
```

Stufenweise Ausführung

```
: gen-innerproduct ( a[row][*] -- xt )
\ xt is of type ( b[*][column] -- n )
>r :noname r>
0 ]] literal SWAP
[[ row-size 0 do ]]
  dup @
  [[ dup @ ]] literal * under+ cell+
  [[ row-byte-size + loop ]]
drop ;
[[ drop ;

: innerproduct ( a[row][*] b[*][column] -- int)
0 row-size 0 do
  >r over @ over @ * r> + >r
  swap cell+ swap row-byte-size +
  r>
loop
>r 2drop r> ;
```

Zusammenfassung

- Mehrere Stacks
- Locals
- Implizite Parameter
- Aufteilung auf Tasks
- Stufenweise Ausführung